

Título del TFM



Facultad de Matemáticas
Departamento de Ciencias de la Computación e Inteligencia Artificial
Trabajo Fin de Máster

Autor

Agradecimientos

El presente Trabajo Fin de Máster se ha realizado en el Departamento de Ciencias de la Computación e Inteligencia Artificial de la Universidad de Sevilla.

Supervisado por

Tutor

Abstract

Resumen en inglés

Esta obra está bajo una licencia Reconocimiento–NoComercial–CompartirIgual 2.5 Spain de Creative Commons.

Se permite:

- copiar, distribuir y comunicar públicamente la obra
- hacer obras derivadas

Bajo las condiciones siguientes:



Reconocimiento. Debe reconocer los créditos de la obra de la manera especificada por el autor.



No comercial. No puede utilizar esta obra para fines comerciales.



Compartir bajo la misma licencia. Si altera o transforma esta obra, o genera una obra derivada, sólo puede distribuir la obra generada bajo una licencia idéntica a ésta.

- Al reutilizar o distribuir la obra, tiene que dejar bien claro los términos de la licencia de esta obra.
- alguna de estas condiciones puede no aplicarse si se obtiene el permiso del titular de los derechos de autor.

Esto es un resumen del texto legal (la licencia completa). Para ver una copia de esta licencia, visite <http://creativecommons.org/licenses/by-nc-sa/2.5/es/> o envíe una carta a Creative Commons, 559 Nathan Abbott Way, Stanford, California 94305, USA.

Índice general

1	Introducción	9
1.1	Introducción a Haskell	9
2	Interpretación algebraica de la lógica	11
2.1	Lógica proposicional	12
2.1.1	Alfabeto y sintaxis	12
2.1.2	Semántica	16
2.1.3	Validez, satisfacibilidad e insatisfacibilidad	18
2.1.4	Bases de conocimiento	19
2.1.5	Consistencia e inconsistencia	20
2.1.6	Consecuencia lógica	21
2.1.7	Equivalencia	23
2.1.8	Retracción conservativa	24
2.2	El anillo $\mathbb{F}_2[x]$	24
2.2.1	Polinomios en Haskell	25
2.2.2	Introducción a HaskellForMaths	28
2.2.3	$\mathbb{F}_2[x]$ en Haskell	32
2.2.4	Transformaciones entre fórmulas y polinomios	36
2.2.5	Correspondencia entre valoraciones y puntos en \mathbb{F}_2^n	39
2.2.6	Proyección polinomial	40
2.2.7	Bases de conocimiento e ideales	42
3	Regla de independencia y prueba no clausal de teoremas	47
3.1	Retracción conservativa mediante omisión de variables	48
3.2	Derivadas Booleanas	53
3.3	Regla de independencia	54
3.4	Cálculo lógico	54
4	Aplicaciones	55
4.1	Introducción a Haskell	55

5 Conclusión	57
5.1 Introducción a Haskell	57
Bibliografía	58
Índice de definiciones	59

Capítulo 1

Introducción

En este capítulo se hace una breve introducción a la programación funcional en Haskell suficiente para entender su aplicación en los siguientes capítulos. Para una introducción más amplia se pueden consultar los apuntes de la asignatura de Informática de 1º del Grado en Matemáticas ([2]).

El contenido de este capítulo se encuentra en el módulo PFH

```
module PFH where
import Data.List
```

1.1. Introducción a Haskell

En esta sección se introducirán funciones básicas para la programación en Haskell. Como método didáctico, se empleará la definición de funciones ejemplos, así como la redefinición de funciones que Haskell ya tiene predefinidas, con el objetivo de que el lector aprenda *“a montar en bici, montando”*.

A continuación se muestra la definición (cuadrado x) es el cuadrado de x. Por ejemplo, La definición es

```
-- |
-- >>> cuadrado 3
-- 9
-- >>> cuadrado 4
-- 16
cuadrado :: Int -> Int
cuadrado x = x * x
```


Capítulo 2

Interpretación algebraica de la lógica

En este capítulo se estudiarán las principales relaciones entre la lógica proposicional y los polinomios con coeficientes en cuerpos finitos, centrando la atención en \mathbb{F}_2 , el cuerpo finito con dos elementos.

La idea principal que subyace en la interpretación algebraica de la lógica es la de hacerle corresponder a cada fórmula un polinomio de forma que la función valor de verdad inducida por la fórmula se pueda entender como una función polinomial de \mathbb{F}_2 . En otras palabras, se persigue que si la fórmula es verdadera, el valor del polinomio que tiene asociado es 1; mientras que si la fórmula es falsa, el polinomio vale 0.

En la Figura 2.1 (abajo) se muestra una representación gráfica de la relación entre las fórmulas proposicionales y los polinomios de $\mathbb{F}_2[x]$. Destacar que se usa el ideal $\mathbb{I}_2 := (x_1 + x_1^2, \dots, x_n + x_n^2) \subseteq \mathbb{F}_2[x]$ y que $proj$ es la proyección natural sobre el anillo cociente.




Figura 2.1: Relación entre las fórmulas proposicionales y $\mathbb{F}[x]_2$

Destacar que se paralelizará la exposición de la teoría y el desarrollo de las imple-

mentaciones en Haskell. Teniendo en cuenta que el objetivo principal de dichos programas es la obtención de una herramienta eficiente para resolver el problema SAT, tema que se tratará con más detalle en los próximos capítulos.

2.1. Lógica proposicional

En esta sección se introducirán brevemente los principales conceptos de la lógica proposicional, además de fijar la notación que se usará durante todo el trabajo.

```
module Logica where
```

Para conseguir este objetivo se utilizarán las siguientes librerías auxiliares:

```
import Control.Monad ( liftM
                        , liftM2)
import Data.List      ( union
                        , subsequences
                        )
import Test.QuickCheck
import qualified Data.Set as S
```

Antes de describir el lenguaje de la lógica proposicional es importante recordar las definiciones formales de alfabeto y lenguaje:

Definición 2.1.1. Un alfabeto es un conjunto finito de símbolos.

Definición 2.1.2. Un *lenguaje* es un conjunto de cadenas sobre un alfabeto.

Especificando, un lenguaje formal es un lenguaje cuyos símbolos primitivos y reglas para unir esos símbolos están formalmente especificados. Al conjunto de las reglas se lo llama gramática formal o sintaxis. A las cadenas de símbolos que siguen las indicaciones de la gramática se les conoce como fórmulas bien formadas o simplemente fórmulas.

Para algunos lenguajes formales existe además una semántica formal que puede interpretar y dar significado a las fórmulas bien formadas del lenguaje. El lenguaje de la lógica proposicional es un caso particular de lenguaje formal con semántica.

2.1.1. Alfabeto y sintaxis

El alfabeto de la lógica proposicional está formado por tres tipos de elementos: las variables proposicionales, las conectivas lógicas y los símbolos auxiliares.

Definición 2.1.3. Las *variables proposicionales* son un conjunto finito de símbolos proposicionales que, tal y como su propio nombre indica, representan proposiciones. Dichas proposiciones son sentencias que pueden ser declaradas como verdaderas o falsas, es por esto que se dice que las variables proposicionales toman valores discretos (True o False). Es comúnmente aceptado (y así será en este trabajo) llamar al conjunto de las variables ($\mathcal{L} = \{p_1, \dots, p_n\}$) lenguaje proposicional. A la hora de implementar las variables proposicionales se representarán por cadenas:

```
type VarProp = String
```

El conjunto de las fórmulas $Form(\mathcal{L})$ se construye usando las conectivas lógicas estándar:

- Monarias:
 - Negación (\neg) – Constante *true* (\top)
 - Constante *false* ($\perp = \neg\top$)
- Binarias:
 - Conjunción (\wedge) – Condicional o implicación (\rightarrow)
 - Disyunción (\vee) – Bicondicional (\leftrightarrow)

Los símbolos auxiliares con los que se trabajará serán los paréntesis, que se utilizan para indicar precedencia y ya vienen implementados en Haskell.

Finalmente, se define el tipo de dato de las fórmulas proposicionales (FProp) de la siguiente manera:

```
data FProp = T
           | F
           | Atom VarProp
           | Neg FProp
           | Conj FProp FProp
           | Disj FProp FProp
           | Impl FProp FProp
           | Equi FProp FProp
deriving (Eq,Ord)
```

Por razones estéticas además de facilitar el uso de este tipo de dato se declara el procedimiento de escritura de las fórmulas:

```
instance Show FProp where
  show (T)      = "\top"
  show (F)      = "\perp"
```

```

show (Atom x)    = x
show (Neg x)     = "¬" ++ show x
show (Conj x y)  = "(" ++ show x ++ " ∧ " ++ show y ++ ")"
show (Disj x y)  = "(" ++ show x ++ " ∨ " ++ show y ++ ")"
show (Impl x y)  = "(" ++ show x ++ " → " ++ show y ++ ")"
show (Equi x y)  = "(" ++ show x ++ " ↔ " ++ show y ++ ")"

```

Las fórmulas atómicas carecen de una estructura formal más profunda; es decir, son aquellas fórmulas que no contienen conectivas lógicas. En la lógica proposicional, las únicas fórmulas atómicas que aparecen son las variables proposicionales. Por ejemplo:

```

p, q, r :: FProp
p  = Atom "p"
q  = Atom "q"
r  = Atom "r"

```

Combinando las fórmulas atómicas mediante el uso de las conectivas lógicas anteriormente enumeradas obtenemos lo que se denomina como fórmulas compuestas. A continuación, implementaremos las conectivas lógicas como funciones entre fórmulas:

$(\text{no } f)$ es la negación de la fórmula f .

```

no :: FProp -> FProp
no = Neg

```

$(f \vee g)$ es la disyunción de las fórmulas f y g .

```

(∨) :: FProp -> FProp -> FProp
(∨)  = Disj
infixr 5 ∨

```

$(f \wedge g)$ es la conjunción de las fórmulas f y g .

```

(∧) :: FProp -> FProp -> FProp
(∧)  = Conj
infixr 4 ∧

```

$(f \rightarrow g)$ es la implicación de la fórmula f a la fórmula g .

```

(→) :: FProp -> FProp -> FProp
(→)  = Impl
infixr 3 →

```

$(f \leftrightarrow g)$ es la equivalencia entre las fórmulas f y g .

```
(↔) :: FProp -> FProp -> FProp
(↔) = Equi
infixr 2 ↔
```

Durante el desarrollo del trabajo se definirán distintas propiedades sobre las fórmulas proposicionales. Es bien sabido que una ventaja que nos ofrece Haskell a la hora de trabajar es poder definir también dichas propiedades y comprobarlas. Sin embargo, como las fórmulas proposicionales se han definido por el usuario el sistema no es capaz de generarlas automáticamente. Es necesario declarar que `FProp` sea una instancia de `Arbitrary`:

```
instance Arbitrary FProp where
  arbitrary = sized prop
  where
    prop n | n <= 0      = atom
           | otherwise   = oneof [
              atom
            , liftM Neg subform
            , liftM2 Conj subform subform
            , liftM2 Disj subform subform
            , liftM2 Impl subform subform
            , liftM2 Equi subform' subform' ]
    where
      atom      = oneof [liftM Atom (elements ["p","q","r","s"]),
                        elements [F,T]]
      subform   = prop (n `div` 2)
      subform'  = prop (n `div` 4)
```

Dadas dos fórmulas F, G y p una variable proposicional, se denota por $F\{p/G\}$ a la fórmula obtenida al sustituir cada ocurrencia de p en F por la fórmula G .

Se implementa $f\{p/g\}$ en la función `(sustituye f p g)`, donde f es la fórmula original, p la variable proposicional a sustituir y g la fórmula proposicional por la que se sustituye:

```
-- | Por ejemplo,
-- >>> sustituye (no p) "p" q
-- ¬q
-- >>> sustituye (no (q ∧ no p)) "p" (q ↔ p)
-- ¬(q ∧ ¬(q ↔ p))
sustituye :: FProp -> VarProp -> FProp -> FProp
sustituye T _ _ = T
```

```

sustituye F          _ _ = F
sustituye (Atom f)   p g | f == p = g
                      | otherwise = Atom f
sustituye (Neg f)     p g = Neg (sustituye f p g)
sustituye (Conj f1 f2) p g = Conj (sustituye f1 p g) (sustituye f2 p g)
sustituye (Disj f1 f2) p g = Disj (sustituye f1 p g) (sustituye f2 p g)
sustituye (Impl f1 f2) p g = Impl (sustituye f1 p g) (sustituye f2 p g)
sustituye (Equi f1 f2) p g = Equi (sustituye f1 p g) (sustituye f2 p g)

```

2.1.2. Semántica

Definición 2.1.4. Una interpretación o valoración es una función $i : \mathcal{L} \rightarrow \{0,1\}$.

En Haskell se representará como un conjunto de fórmulas atómicas. Las fórmulas que aparecen en dicho conjunto se suponen verdaderas, mientras que las restantes fórmulas atómicas se suponen falsas.

```
type Interpretacion = [FProp]
```

El significado de la fórmula f en la interpretación i viene dado por la función de verdad en su sentido más clásico, tal y como se detalla en el código.

(significado f i) es el significado de la fórmula f en la interpretación i :

```

-- | Por ejemplo,
--
-- >>> significado ((p ∨ q) ∧ ((no q) ∨ r)) [r]
-- False
-- >>> significado ((p ∨ q) ∧ ((no q) ∨ r)) [p,r]
-- True
significado :: FProp -> Interpretacion -> Bool
significado T          _ = True
significado F          _ = False
significado (Atom f)   i = (Atom f) 'elem' i
significado (Neg f)     i = not (significado f i)
significado (Conj f g) i = (significado f i) && (significado g i)
significado (Disj f g) i = (significado f i) || (significado g i)
significado (Impl f g) i = significado (Disj (Neg f) g) i
significado (Equi f g) i = significado (Conj (Impl f g) (Impl g f)) i

```

Una interpretación i es modelo de la fórmula $F \in \text{Form}(\mathcal{L})$ si hace verdadera la fórmula en el sentido clásico definido anteriormente. La función (esModeloFormula i f) se verifica si la interpretación i es un modelo de la fórmula f .


```
-- | Por ejemplo,
--
-- >>> esModeloFormula [r] ((p ∨ q) ∧ ((no q) ∨ r))
-- False
-- >>> esModeloFormula [p,r] ((p ∨ q) ∧ ((no q) ∨ r))
-- True
esModeloFormula :: Interpretacion -> FProp -> Bool
esModeloFormula i f = significado f i
```

Se denota por $Mod(F)$ al conjunto de modelos de F . Para implementarlo se necesitan dos funciones auxiliares.

$(\text{simbolosPropForm } f)$ es el conjunto formado por todos los símbolos proposicionales que aparecen en la fórmula f .

```
-- | Por ejemplo,
--
-- >>> simbolosPropForm (p ∧ q → p)
-- [p,q]
simbolosPropForm :: FProp -> [FProp]
simbolosPropForm T      = []
simbolosPropForm F      = []
simbolosPropForm (Atom f) = [(Atom f)]
simbolosPropForm (Neg f)  = simbolosPropForm f
simbolosPropForm (Conj f g) = simbolosPropForm f 'union' simbolosPropForm g
simbolosPropForm (Disj f g) = simbolosPropForm f 'union' simbolosPropForm g
simbolosPropForm (Impl f g) = simbolosPropForm f 'union' simbolosPropForm g
simbolosPropForm (Equi f g) = simbolosPropForm f 'union' simbolosPropForm g
```

$(\text{interpretacionesForm } f)$ es la lista de todas las interpretaciones de la fórmula f .

```
-- | Por ejemplo,
--
-- >>> interpretacionesForm (p ∧ q → p)
-- [[], [p], [q], [p,q]]
interpretacionesForm :: FProp -> [Interpretacion]
interpretacionesForm f = subsequences (simbolosPropForm f)
```

$(\text{modelosFormula } f)$ es la lista de todas las interpretaciones de la fórmula f que son modelo de la misma.

```
-- | Por ejemplo,
--
```

```
-- >>> modelosFormula ((p ∨ q) ∧ ((no q) ∨ r))
-- [[p],[p,r],[q,r],[p,q,r]]
modelosFormula :: FProp -> [Interpretacion]
modelosFormula f = [i | i <- interpretacionesForm f, esModeloFormula i f]
```

2.1.3. Validez, satisfacibilidad e insatisfacibilidad

Definición 2.1.5. Una fórmula F se dice válida si toda interpretación i de F es modelo de la fórmula. La función (`esValida f`) se verifica si la fórmula f es válida.

```
-- | Por ejemplo,
--
-- >>> esValida (p → p)
-- True
-- >>> esValida (p → q)
-- False
-- >>> esValida ((p → q) ∨ (q → p))
-- True
esValida :: FProp -> Bool
esValida f = modelosFormula f == interpretacionesForm f
```

Definición 2.1.6. Una fórmula F se dice insatisfacible si no existe ninguna interpretación i de F que sea modelo de la fórmula. La función (`esInsatisfacible f`) se verifica si la fórmula f es insatisfacible.

```
-- | Por ejemplo,
--
-- >>> esInsatisfacible (p ∧ (no p))
-- True
-- >>> esInsatisfacible ((p → q) ∧ (q → r))
-- False
esInsatisfacible :: FProp -> Bool
esInsatisfacible f = modelosFormula f == []
```

Definición 2.1.7. Una fórmula F se dice satisfacible si existe al menos una interpretación i de F que sea modelo de la fórmula. La función (`esSatisfacible f`) se verifica si la fórmula f es satisfacible.

```
-- | Por ejemplo,
--
-- >>> esSatisfacible (p ∧ (no p))
-- False
```

```
-- >>> esSatisfacible ((p → q) ∧ (q → r))
-- True
esSatisfacible :: FProp -> Bool
esSatisfacible = not . null . modelosFormula
```

2.1.4. Bases de conocimiento

Definición 2.1.8. Una *base de conocimiento* o *Knowledge Basis* (*KB*) es un conjunto finito de fórmulas proposicionales. Se define el tipo de dato *KB* como:

```
type KB = S.Set FProp
```

Destacar que se denotará al lenguaje proposicional de K como $\mathcal{L}(K)$.

Antes de extender las definiciones anteriores a más de una fórmula, se implementarán dos funciones auxiliares. El objetivo de dichas funciones es obtener toda la ca suística de interpretaciones posibles de un conjunto de fórmulas o *KB*.

(simbolosPropKB *k*) es el conjunto de los símbolos proposicionales de la base de conocimiento *k*.

```
-- | Por ejemplo,
--
-- >>> simbolosPropKB (S.fromList [p ∧ q → r, p → r])
-- [p,r,q]
simbolosPropKB :: KB -> [FProp]
simbolosPropKB = foldl (\acc f -> union acc (simbolosPropForm f)) []
```

(interpretacionesKB *k*) es la lista de las interpretaciones de la base de conocimiento *k*.

```
-- | Por ejemplo,
--
-- >>> interpretacionesKB (S.fromList [p → q, q → r])
-- [[], [p], [q], [p,q], [r], [p,r], [q,r], [p,q,r]]
interpretacionesKB :: KB -> [Interpretacion]
interpretacionesKB = subsequences . simbolosPropKB
```

Definición 2.1.9. Análogamente al caso de una única fórmula, se dice que *i* es *modelo* de *K* ($i \models K$) si lo es de cada una de las fórmulas de *K*. La función `texttt(esModeloKB i k)` se verifica si la interpretación *i* es un modelo de todas las fórmulas de la base de conocimiento *k*.

```
-- | Por ejemplo,
--
-- >>> esModeloKB [r] (S.fromList [q,no p ,r])
-- False
-- >>> esModeloKB [q,r] (S.fromList [q,no p ,r])
-- True
esModeloKB :: Interpretacion -> KB -> Bool
esModeloKB i = all (esModeloFormula i)
```

Al conjunto de modelos de K se le denota por $Mod(K)$. $(modelosKB\ k)$ es la lista de modelos de la base de conocimiento k .

```
-- | Por ejemplo,
--
-- >>> modelosKB $ S.fromList [(p ∨ q) ∧ ((no q) ∨ r), q → r]
-- [[p],[p,r],[q,r],[p,q,r]]
-- >>> modelosKB $ S.fromList [(p ∨ q) ∧ ((no q) ∨ r), r → q]
-- [[p],[q,r],[p,q,r]]
modelosKB :: KB -> [Interpretacion]
modelosKB s = [i | i <- interpretacionesKB s, esModeloKB i s]
```

2.1.5. Consistencia e inconsistencia

La consistencia es una propiedad de las bases de conocimiento que se puede definir de dos maneras distintas:

Definición 2.1.10. Un conjunto de fórmulas se dice *consistente* si y sólo si tiene al menos un modelo. Una definición alternativa es que dicho conjunto de fórmulas es *consistente* si y sólo si para toda fórmula f no es posible deducir tanto f como $\neg f$ a partir de él.

La función $(esConsistente\ k)$ se verifica si la base de conocimiento k es consistente.

```
-- |Por ejemplo,
--
-- >>> esConsistente $ S.fromList [(p ∨ q) ∧ ((no q) ∨ r), p → r]
-- True
-- >>> esConsistente $ S.fromList [(p ∨ q) ∧ ((no q) ∨ r), p → r, no r]
-- False
esConsistente :: KB -> Bool
esConsistente = not . null . modelosKB
```

Definición 2.1.11. Un conjunto de fórmulas se dice *inconsistente* si y sólo si no tiene modelo. Una definición alternativa es que dicho conjunto de fórmulas es *consistente* si y sólo si alguna fórmula f es posible deducir tanto f como $\neg f$ a partir de él.

La función (`esInconsistente k`) se verifica si la base de conocimiento k es inconsistente.

```
-- |Por ejemplo,
--
-- >>> esInconsistente $ S.fromList [(p ∨ q) ∧ ((no q) ∨ r), p → r]
-- False
-- >>> esInconsistente $ S.fromList [(p ∨ q) ∧ ((no q) ∨ r), p → r, no r]
-- True
esInconsistente :: KB -> Bool
esInconsistente = null . modelosKB
```

2.1.6. Consecuencia lógica

La consecuencia lógica es la relación entre las premisas y la conclusión de lo que se conoce como un argumento deductivamente válido. Esta relación es un concepto fundamental en la lógica y aparecerá con asiduedad en el desarrollo del trabajo.

Existe una manera de caracterizar la relación de consecuencia lógica basada en los axiomas y las reglas de inferencia. Sin embargo, se abordará la definición desde otra perspectiva, teniendo en cuenta su implementación.

Definición 2.1.12. Se dice que F es *consecuencia lógica* de K ($K \models F$) si todo modelo de K lo es a su vez de F , es decir, $\text{Mod}(K) \subseteq \text{Mod}(F)$. Equivalentemente, $K \models F$ si no es posible que las premisas sean verdaderas y la conclusión falsa.

La función (`esConsecuencia k f`) se verifica si la fórmula proposicional f es consecuencia lógica de la base de conocimiento o conjunto de fórmulas k .

```
-- |Por ejemplo,
--
-- >>> esConsecuencia (S.fromList [p → q, q → r]) (p → r)
-- True
-- >>> esConsecuencia (S.fromList [p]) (p ∧ q)
-- False
esConsecuencia :: KB -> FProp -> Bool
esConsecuencia k f =
  null [i | i <- interpretacionesKB (S.insert f k)
        , esModeloKB i k
        , not (esModeloFormula i f)]
```

Con el objetivo de hacer más robusto el sistema se implementarán dos propiedades de la relación de *ser consecuencia* en lógica proposicional. Dichas propiedades se comprobarán con la librería QuickCheck propia del lenguaje Haskell. Es importante saber que estas comprobaciones no son más que meros chequeos de que una propiedad se cumple para una batería de ejemplos. En ningún caso se puede confiar en que dicha propiedad se cumple el 100% del tiempo y en ningún caso trivial.

Proposición 2.1.13. Una fórmula f es válida si y sólo si es consecuencia del conjunto vacío.

```
-- |
-- >>> quickCheck prop_esValida
-- +++ OK, passed 100 tests.
prop_esValida :: FProp -> Bool
prop_esValida f =
    esValida f == esConsecuencia S.empty f
```

Proposición 2.1.14. Una fórmula f es consecuencia de un conjunto de fórmulas k si y sólo si dicho el conjunto formado por k y $\neg f$ es inconsistente.

```
-- |
-- >>> quickCheck prop_esConsecuencia
-- +++ OK, passed 100 tests.
prop_esConsecuencia :: KB -> FProp -> Bool
prop_esConsecuencia k f =
    esConsecuencia k f == esInconsistente (S.insert (Neg f) k)
```

De manera natural se extiende la definición de *ser consecuencia* a bases de conocimiento en lugar de fórmulas, preservando la misma notación. La función (`esConsecuenciaKB k k'`) se verifica si todas las fórmulas del conjunto k' son consecuencia de las del conjunto k .

```
-- |Por ejemplo,
--
-- >>> esConsecuenciaKB (S.fromList [p -> q, q -> r]) (S.fromList [p -> q, p -> r])
-- True
-- >>> esConsecuenciaKB (S.fromList [p]) (S.fromList [p ^ q])
-- False
esConsecuenciaKB :: KB -> KB -> Bool
esConsecuenciaKB k = all (esConsecuencia k)
```

2.1.7. Equivalencia

Definición 2.1.15. Sean F y G dos fórmulas proposicionales, se dice que son equivalentes ($F \equiv G$) si tienen el mismo contenido lógico, es decir, si tienen el mismo valor de verdad en todas sus interpretaciones.

La función (`equivalentes f g`) se verifica si las fórmulas proposicionales son equivalentes.

```
-- |Por ejemplo,
--
-- >>> equivalentes (p → q) (no p ∨ q)
-- True
-- >>> equivalentes (p) (no (no p))
-- True
equivalentes :: FProp -> FProp -> Bool
equivalentes f g = esValida (f ↔ g)
```

Definición 2.1.16. Dadas K y K' dos bases de conocimiento, se dice que son equivalentes ($K \equiv K'$) si $K \models K'$ y $K' \models K$.

La función (`equivalentesKB k k'`) se verifica si las bases de conocimiento k y k' son equivalentes.

```
-- |Por ejemplo,
--
-- >>> equivalentesKB (S.fromList [p → q, r ∨ q]) (S.fromList [no p ∨ q, q ∨ r])
-- True
-- >>> equivalentesKB (S.fromList [p ∧ q]) (S.fromList [q, p])
-- True
equivalentesKB :: KB -> KB -> Bool
equivalentesKB k k' = esConsecuenciaKB k k' && esConsecuenciaKB k' k
```

La siguiente propiedad comprueba si las definiciones implementadas anteriormente son iguales en el caso de una única fórmula en la base de conocimiento.

Proposición 2.1.17. Sean F y g dos fórmulas proposicionales, son equivalentes como fórmulas si y sólo si lo son como bases de conocimiento.

```
-- |
-- >>> quickCheck prop_equivalentes
-- +++ OK, passed 100 tests.
prop_equivalentes :: FProp -> FProp -> Bool
prop_equivalentes f g =
  equivalentes f g == equivalentesKB (S.singleton f) (S.singleton g)
```

2.1.8. Retracción conservativa

Dada una base de conocimiento resulta interesante estudiar qué fórmulas se pueden deducir de la misma o incluso buscar si existe otra manera de expresar el conocimiento, en definitiva otras fórmulas, de las que se deduzca exactamente la misma información. A esta relación entre bases de conocimiento se le conoce con el nombre de extensión:

Definición 2.1.18. Sean K y K' bases de conocimiento, se dice que K es una *extensión* de K' si $\mathcal{L}(K') \subseteq \mathcal{L}(K)$ y

$$\forall F \in \text{Form}(\mathcal{L}(K')) \quad [K' \models F \Rightarrow K \models F]$$

Si nos restringimos a las fórmulas consecuencia de una base de conocimiento en el lenguaje de la otra:

Definición 2.1.19. Sean K y K' bases de conocimiento, se dice que K es una *extensión conservativa* de K' si es una extensión tal que toda consecuencia lógica de K expresada en el lenguaje $\mathcal{L}(K')$, es también consecuencia de K' ,

$$\forall F \in \text{Form}(\mathcal{L}(K')) \quad [K \models F \Rightarrow K' \models F]$$

es decir, K extiende a K' pero no se añade nueva información expresada en términos de $\mathcal{L}(K')$.

Definición 2.1.20. K' es una *retracción conservativa* de K si y sólo si K es una extensión conservativa de K' .

Dado $\mathcal{L}' \subseteq \mathcal{L}(K)$, siempre existe una retracción conservativa de K al lenguaje \mathcal{L}' . La *retracción conservativa canónica* de K a \mathcal{L}' se define como:

$$[K, \mathcal{L}'] = \{F \in \text{Form}(\mathcal{L}') : K \models F\}$$

Esto es, $[K, \mathcal{L}']$ es el conjunto de \mathcal{L}' -fórmulas que son consecuencia de K . De hecho, cualquier retracción conservativa sobre \mathcal{L}' es equivalente a $[K, \mathcal{L}']$. El problema real es dar una axiomatización de dicho conjunto de fórmulas.

2.2. El anillo $\mathbb{F}_2[\mathbf{x}]$

Para una natural interpretación algebraica de la lógica, el marco de trabajo escogido es el anillo $\mathbb{F}_2[\mathbf{x}]$. Con la idea de facilitar la identificación entre variables proposicionales e incógnitas, se fija la notación de forma que a una variable proposicional p_i (ó p)

le corresponde la incógnita x_i (ó x_p).

La notación usada en los polinomios es la estándar. Dado $\alpha = (\alpha_1, \dots, \alpha_n) \in \mathbb{N}^n$, se define $|\alpha| := \max\{\alpha_1, \dots, \alpha_n\}$ y x^α es el monomio $x_1^{\alpha_1} \dots x_n^{\alpha_n}$.

Definición 2.2.1. El *grado* de $a(\mathbf{x}) \in \mathbb{F}_2[\mathbf{x}]$ es

$$\deg_\infty(a(\mathbf{x})) := \max\{|\alpha| : \mathbf{x}^\alpha \text{ es un monomio de } a\}$$

Si $\deg(a(x)) \leq 1$, el polinomio $a(\mathbf{x})$ se denomina *fórmula polinomial*. El grado de $a(x)$ respecto una variable x_i se denota por $\deg_i(a(\mathbf{x}))$.

A continuación se tratará de implementar $\mathbb{F}_2[\mathbf{x}]$ en Haskell, así como algunas operaciones polinomiales que serán necesarias más adelante.

2.2.1. Polinomios en Haskell

Si se quiere trabajar con polinomios en Haskell, no es necesario “hacer tabla rasa” y tratar de implementarlo todo desde el principio. Parafraseando a Newton,

Si he conseguido ver más lejos es porque me he aupado en hombros de gigantes
Isaac Newton

así que conviene apoyarse en la multitud de librerías existentes de la comunidad Haskell. Aunque, continuando con la metáfora de Newton, no es fácil saber qué gigante es el más alto si miramos desde el suelo. Por tanto, es necesario un estudio de las distintas librerías, a fin de escoger la que se adecúe en mayor medida a las necesidades de este proyecto.

Seguidamente se comentarán los detalles más relevantes de las distintas librerías estudiadas, centrando la atención en los detalles prácticos como la empleabilidad y la eficiencia.

Cryptol

Cryptol es un lenguaje de programación especialmente desarrollado para implementar algoritmos criptográficos. Su ventaja es su similitud con el lenguaje matemático respecto a un lenguaje de propósito general. Está escrito en Haskell, siendo un trabajo muy pulido con un tutorial muy completo.

Sin embargo, no resulta intuitivo aislar la librería de polinomios por lo que para realizar las pruebas se ha optado por cargar todo el paquete apoyándonos en la documentación.

Esta librería sólo trabaja con polinomios en una única variable. Por ejemplo, los polinomios de $\mathbb{F}_2[x]$ con grado menor o igual que 7 se codifican mediante un vector de unos y ceros donde se almacenan los coeficientes de cada x^i con $i \leq 7$. Debido a este tipo de codificación no es trivial extenderlo a más de una variable.

The polynomial package

El módulo `MATh.Polynomial` de esta librería implementa la aritmética en una única variable, debiendo especificar con qué orden monomial se quiere trabajar. Por el contrario, no es necesario dar como entrada el cuerpo en el que se defina la K -álgebra sino que construye los polinomios directamente de una lista ordenada de coeficientes. Por tanto, hereda el tipo de dichos elementos de la lista (`[a]`), formando lo que se denomina como tipo polinomio (`Poly a`).

La librería se centra en definir distintos tipos de polinomios de la literatura matemática clásica como los polinomios de Hermite, de Bernouilli, las bases de Newton o las bases de Lagrange. Finalmente, destacar que la documentación no es escasa por lo que su uso resulta tedioso.

ToySolver.Data.Polynomial

Este módulo se enmarca en una librería que trata de implementar distintos procedimientos y algoritmos para resolver varios problemas de decisión. Debido a que su función es únicamente servir como auxiliar para otros módulos el código no está comentado, lo que dificulta su uso.

La implementación es muy completa, incluye multitud de operaciones y procedimientos de polinomios y \mathcal{K} -álgebras. Por ejemplo, permite construir polinomios sobre cuerpos finitos. Sin embargo, la estructura de dato de los polinomios no es intuitiva para su uso.

SBV

El parecido a la librería de `Cryptol` es notable, por lo que se encuentra con inconvenientes similares.

Gröebner bases in Haskell

El objetivo principal de esta librería es, tal y como su nombre indica, el cálculo de bases de Gröebner mediante operaciones con ideales.

En lo que respecta al código, al tipo de dato polinomio se le debe especificar el Anillo de coeficientes así como el orden monomial. Además, se deben declarar desde el principio las variables que se quieren usar.

La documentación está muy detallada aunque el autor comenta que se prioriza la claridad del código y el rigor matemático ante la eficiencia.

HaskellForMaths

El módulo de polinomios es `Math.CommutativeAlgebra.Polynomial`. Al igual que en la librería anterior, se permite escoger el orden monomial entre los tres más usuales (lexicográfico, graduado lexicográfico y graduado reverso lexicográfico) e incluso definir otros nuevos.

También se debe dar como entrada el cuerpo sobre el que se trabajará. Para ello se incorpora un módulo específico llamado `Math.Core.Field` en el que están implementados los cuerpos más comunes, como los números racionales o los cuerpos finitos.

Destacar que el tipo de dato polinomio tiene estructura vectorial, donde la base es cada monomio que exista (combinaciones de todas las variables) y el número en la posición i -ésima del vector corresponde con el coeficiente del monomio i -ésimo (según el orden especificado) del polinomio.

Las operaciones básicas de los polinomios están ya implementadas de forma eficiente, destacando la multiplicación, desarrollada en un módulo aparte se apoya en otra librería de productos tensoriales.

Esta librería responde en líneas generales a las necesidades del proyecto ya que es modular, el código está documentado y la mayoría de algoritmos son eficientes. Por dichas razones se escoge esta librería como auxiliar en el proyecto a desarrollar.

2.2.2. Introducción a HaskellForMaths

Se muestran a continuación diversos ejemplos de funciones de Haskell4Maths que aparecerán de forma recurrente en el resto del trabajo.

```
module Haskell4Maths
  ( F2
  , Vect(..)
  , linear
  , zeroV
  , Lex
  , Glex
  , Grevlex
  , var
  , mindices
  , lm
  , lt
  , eval
  , (%%)
  , vars) where

import Math.Core.Field
import Math.Algebras.VectorSpace
import Math.CommutativeAlgebra.Polynomial
```

Math.Core.Field

En este módulo se definen el cuerpo \mathbb{Q} de los racionales y los cuerpos finitos o de Galois: $\mathbb{F}_2, \mathbb{F}_3, \mathbb{F}_4, \mathbb{F}_5, \mathbb{F}_7, \mathbb{F}_8, \mathbb{F}_9, \mathbb{F}_{11}, \mathbb{F}_{13}, \mathbb{F}_{16}, \mathbb{F}_{17}, \mathbb{F}_{19}, \mathbb{F}_{23}, \mathbb{F}_{25}$.

Veamos unos ejemplos de cómo se trabaja con los números racionales:

```
-- |
-- >>> (7/14 :: Q)
-- 1/2
-- >>> (0.6 :: Q)
-- 3/5
-- >>> (2.3 + 1/5 * 4/7) :: Q
-- 169/70
```

Para este trabajo, el cuerpo que nos interesa es \mathbb{F}_2 , cuyos elementos pertenecen a la lista f2:

```
-- |
-- >>> f2
-- [0,1]
```

Y cuyas operaciones aritméticas se definen de forma natural:

```
-- |
-- >>> (2 :: F2)
-- 0
-- >>> (1 :: F2) + (3 :: F2)
-- 0
-- >>> (7 :: F2) * (1 :: F2)
-- 1
```

Además, cuenta con la función auxiliar `fromInteger` para transformar números de tipo `Integer` en el tipo `Fp` dónde `p` es número de elementos del cuerpo:

```
-- |
-- >>> (fromInteger (12345 :: Integer)) :: F2
-- 1
```

Math.Algebras.VectorSpace

En este módulo se define el tipo y las operaciones de los espacios de k -vectores libres sobre una base b , con k un cuerpo, de la siguiente manera:

```
newtype Vect k b = V [(b,k)]
```

Intuitivamente, un vector es una lista de pares donde la primera coordenada es un elemento de la base y la segunda coordenada el coeficiente de dicho elemento. Notar que el coeficiente pertenece al cuerpo k , que se debe especificar en el tipo.

También destacar que este nuevo tipo tiene las instancias `Eq`, `Ord` y `Show` además de estar definida la suma y la multiplicación de vectores (en función de la base y los coeficientes).

La función `(zerov)` representa al vector cero independientemente del cuerpo k y la base b . Por ejemplo,

```
-- |
-- >>> zerov :: (Vect Q [a])
```

```
-- 0
-- >>> zeroV :: (Vect F2 F3)
-- 0
```

Una función que conviene destacar es la función `(linear f v)`, que es un mapeo lineal entre dos espacios vectoriales ($A = \text{Vect } k \text{ a}$ y $B = \text{Vect } k \text{ b}$). La función `f :: a -> Vect k b` va de los elementos de la base de A a B . Por lo que `(linear)` es muy útil si se necesita transformar vectores de forma interna.

Math.CommutativeAlgebra.Polynomial

En el siguiente módulo se define el álgebra conmutativa de los polinomios sobre el cuerpo k . Los polinomios se representan como el espacio de k -vectores libres con los monomios como su base.

Para poder trabajar con los polinomios es necesario especificar un orden monomial. En este módulo están implementados los tres más comunes: el lexicográfico (`Lex`), el graduado lexicográfico (`Glex`) y el graduado reverso lexicográfico (`Grevlex`). Asimismo, es posible añadir otros nuevos en caso de que fuera necesario.

La función `(var v)` crea una variable en espacio vectorial de polinomios. Por ejemplo, si se quiere trabajar en $\mathbb{Q}[x, y, z]$, debemos definir:

```
-- |
-- >>> [x,y,z] = map var ["x","y","z"] :: [GlexPoly Q String]
```

Destacar que, en general, es necesario proporcionar los tipos de datos de forma que el compilador sepa qué cuerpo y qué orden monomial usar. A continuación se mostrarán diversos ejemplos de operaciones entre polinomios, variando el orden monomial y el cuerpo.

```
-- |
-- >>> [x,y,z] = map var ["x","y","z"] :: [LexPoly Q String]
-- >>> x^2+x*y+x*z+x*y^2+y*z+y*z^2+z+1
-- x^2+xy+xz+x*y^2+yz+y*z^2+z+1
-- >>> [x,y,z] = map var ["x","y","z"] :: [GlexPoly Q String]
-- >>> x^2+x*y+x*z+x*y^2+y*z+y*z^2+z+1
-- x^2+xy+xz+y^2+yz+z^2+xy+z+1
-- >>> [x,y,z] = map var ["x","y","z"] :: [GrevlexPoly Q String]
-- >>> x^2+x*y+x*z+x*y^2+y*z+y*z^2+z+1
-- x^2+xy+y^2+xz+yz+z^2+xy+z+1
-- >>> [x,y,z] = map var ["x","y","z"] :: [LexPoly Q String]
-- >>> (x+y+z)^2
-- x^2+2xy+2xz+y^2+2yz+z^2
```

```
-- >>> [x,y,z] = map var ["x","y","z"] :: [LexPoly F2 String]
-- >>> (x+y+z)^2
-- x^2+y^2+z^2
```

Como se mencionó anteriormente la base del espacio vectorial que es un polinomio, está formada por monomios. El tipo de dato monomio está formado por un coeficiente i y una lista de pares. En el caso de los polinomios, un ejemplo de monomio es:

```
-- |
-- >>> monomio
-- x^2y
monomio :: MonImpl [Char]
monomio = (M 1 [("x",2),("y",1)])
```

Dichos pares se obtienen mediante la función (`mindices m`) y se pueden entender como los elementos canónicos que forman cada monomio de la base, así como su exponente:

```
-- |
-- >>> mindices monomio
-- [("x",2),("y",1)]
```

En este módulo también se implementan tres funciones auxiliares que resultarán de gran utilidad más adelante. La función auxiliar que resulta más natural implementar cuando se trabaja con polinomios y que además goza de una gran importancia es (`vars p`). Ésta devuelve la lista de variables que aparecen en el polinomio p .

```
-- |Por ejemplo,
--
-- >>> [x,y,z] = map var ["x","y","z"] :: [LexPoly F2 String]
-- >>> vars (x*z*y+y*x^2+z^4)
-- [x,y,z]
```

La segunda función auxiliar es (`lt m`) (término líder) que devuelve un par (m,i) donde m es el monomio líder e i su coeficiente ($i \in k$).

```
-- |
-- >>> [x,y,z] = map var ["x","y","z"] :: [LexPoly F2 String]
-- >>> lt (x*z*y+y*x^2+z^4)
-- (x^2y,1)
```

La tercera es la función (`lm p`) que devuelve el monomio líder del polinomio p :

```
-- |
-- >>> [x,y,z] = map var ["x","y","z"] :: [LexPoly F2 String]
-- >>> lm (x*z*y+y*x^2+z^4)
-- x^2y
```

Otra función natural es `(eval p vs)`, que evalúa el polinomio p en el punto descrito por vs , siendo ésta una lista de pares variable-valor.

```
-- |
-- >>> [x,y] = map var ["x","y"] :: [LexPoly F2 String]
-- >>> eval (x^2+y^2) [(x,1),(y,0)]
-- 1
```

Por último, destacar la función `(p %% xs)` que calcula la reducción del polinomio p respecto de los polinomios de la lista xs .

```
-- |
-- >>> [x,y,z] = map var ["x","y","z"] :: [LexPoly F2 String]
-- >>> (x^2+y^2) %% [x^2+1]
-- y^2+1
```

2.2.3. $\mathbb{F}_2[x]$ en Haskell

En esta subsección se realizarán las implementaciones necesarias para poder trabajar en Haskell con $\mathbb{F}_2[x]$.

```
{-# LANGUAGE FlexibleInstances, FlexibleContexts #-}
module F2 ( VarF2
           , PolF2
           , unbox ) where

import Haskell4Maths ( Vect
                      , Lex
                      , F2
                      , var)
import Test.QuickCheck ( Arbitrary
                        , Gen
                        , arbitrary
                        , vectorOf
                        , choose
                        , quickCheck)
```


El primer paso tras el análisis realizado sobre la librería `HaskellForMaths` en el apartado anterior es definir el tipo de dato que representa $\mathbb{F}_2[x]$ (`PolF2`), así como sus variables (`VarF2`):

```
newtype VarF2 = Box (Vect F2 (Lex String))
    deriving (Eq, Ord)

type PolF2 = Vect F2 (Lex String)
```

Notar que el tipo de las variables es simplemente un cambio de nombre respecto a los polinomios que ha sido metido dentro del constructor `Box`. Este artificio es necesario ya que no se pueden declarar instancias (como se hará a continuación) repetidas sobre un mismo tipo de dato aunque tengan nombres distintos.

Sin embargo, es necesario definir la función auxiliar (`unbox x`) que saca a x de la mónada `Var`:

```
unbox :: VarF2 -> PolF2
unbox (Box x) = x
```

Para poder mostrar por consola las variables de forma estética; es decir, sin mostrar el constructor `Box`, declaramos la instancia `Show`:

```
instance Show VarF2 where
    show = show . unbox
```

Para poder definir propiedades que involucren a estos tipos de datos y comprobarlas con `QuickCheck` es necesario añadir la instancia `Arbitrary`, así como definir generadores de dichos tipos. Se comenzará por el tipo `VarF2` ya que servirá como apoyo para el de los polinomios:

```
instance Arbitrary VarF2 where
    arbitrary = varGen
```

La función `varGen` es un generador de variables:

```
varGen :: Gen VarF2
varGen = do
    n <- choose ((1::Int),100)
    return (Box (var ('x':(show n))))
```

Se declara la instancia Arbitrary para el tipo de dato de los polinomios:

```
instance Arbitrary PolF2 where
  arbitrary = polGen
```

El generador aleatorio de polinomios seguirá la siguiente estructura: en primer lugar se generarán aleatoriamente pares de variable-exponente, con los que se formarán monomios. A partir de la suma de éstos se obtendrán los polinomios:

```
varExpGen :: Gen (PolF2,Int)
varExpGen = do
  Box x <- varGen
  i <- choose ((1::Int),5)
  return $ (x,i)

monGen :: Gen PolF2
monGen = do
  n <- choose ((1::Int),5)
  xs <- vectorOf n varExpGen
  return $ product [ x ^ i | (x,i) <- xs]

polGen :: Gen PolF2
polGen = do
  n <- choose ((1::Int),5)
  xs <- vectorOf n monGen
  return $ sum xs
```

Propiedades de $\mathbb{F}_2[x]$

Es importante comprobar que el nuevo tipo de dato que hemos definido cumple las propiedades básicas. Ya que el trabajo se basa en este tipo de dato y sus propiedades. Se comprobarán las propiedades de la suma y del producto de polinomios de $\mathbb{F}_2[x]$:

La suma de polinomios es conmutativa, $\forall p, q \in \mathbb{F}_2[x] (p + q = q + p)$.

```
-- |
-- >>> quickCheck prop_suma_conmutativa
-- +++ OK, passed 100 tests.
prop_suma_conmutativa :: PolF2 -> PolF2 -> Bool
prop_suma_conmutativa p q = p+q == q+p
```

La suma de polinomios es asociativa: $\forall p, q, r \in \mathbb{F}_2[x] (p + (q + r) = (p + q) + r)$.

```
-- |
-- >>> quickCheck prop_suma_asociativa
-- +++ OK, passed 100 tests.
prop_suma_asociativa :: PolF2 -> PolF2 -> PolF2 -> Bool
prop_suma_asociativa p q r = p+(q+r) == (p+q)+r
```

El cero es el elemento neutro de la suma de polinomios:

```
-- |
-- >>> quickCheck prop_suma_neutro
-- +++ OK, passed 100 tests.
prop_suma_neutro :: PolF2 -> Bool
prop_suma_neutro p = (p + 0 == p) && (0 + p == p)
```

Todo polinomio es simétrico de sí mismo respecto a la suma: $\forall p \in \mathbb{F}_2[x] : p + p = 0$.

```
-- |
-- >>> quickCheck prop_suma_simetrico
-- +++ OK, passed 100 tests.
prop_suma_simetrico :: PolF2 -> Bool
prop_suma_simetrico p = p+p == 0
```

La multiplicación de polinomios es conmutativa: $\forall p, q \in \mathbb{F}_2[x] (p * q = q * p)$.

```
-- |
-- >>> quickCheck prop_prod_conmutativa
-- +++ OK, passed 100 tests.
prop_prod_conmutativa :: PolF2 -> PolF2 -> Bool
prop_prod_conmutativa p q = p*q == q*p
```

El producto es asociativo: $\forall p, q, r \in \mathbb{F}_2[x] (p * (q * r) = (p * q) * r)$.

```
-- |
-- >>> quickCheck prop_prod_asociativo
-- +++ OK, passed 100 tests.
prop_prod_asociativo :: PolF2 -> PolF2 -> PolF2 -> Bool
prop_prod_asociativo p q r = p*(q*r) == (p*q)*r
```

El 1 es el elemento neutro de la multiplicación de polinomios:

```
-- |
-- >>> quickCheck prop_prod_neutro
-- +++ OK, passed 100 tests.
prop_prod_neutro :: PolF2 -> Bool
prop_prod_neutro p = (p * 1 == p) && (1 * p == p)
```

Distributividad del producto respecto la suma: $\forall p, q, r \in \mathbb{F}_2[x] (p * (q + r) = p * q + p * r)$

```
-- |
-- >>> quickCheck prop_distributiva
-- +++ OK, passed 100 tests.
prop_distributiva :: PolF2 -> PolF2 -> PolF2 -> Bool
prop_distributiva p q r = p*(q+r) == (p*q)+(p*r)
```

2.2.4. Transformaciones entre fórmulas y polinomios

La traducción o transformación de la lógica proposicional en álgebra polinomial viene dada por [4] y se ilustra en la figura 2.1.

La idea principal es que las fórmulas se pueden ver como polinomios sobre fórmulas atómicas cuando éstas están expresadas en términos de las conectivas booleanas *o exclusivo* e *y*; así como de las constantes 1 y 0, que equivalen a los conceptos de *Verdad* y *Falsedad*, respectivamente. Las operaciones básicas de suma y multiplicación se corresponden con las conectivas booleanas *o exclusivo* e *y*, respectivamente.

Por tanto, el mapeo $P : Form(\mathcal{L} \rightarrow \mathbb{F}_2[x])$ que aparece en la página 11 se define por:

- $P(\perp) = 0, P(p_i) = x_i, P(\neg F) = 1 + P(F)$
- $P(F_1 \wedge F_2) = P(F_1) \cdot P(F_2)$
- $P(F_1 \vee F_2) = P(F_1) + P(F_2) + P(F_1) \cdot P(F_2)$
- $P(F_1 \rightarrow F_2) = 1 + P(F_1) + P(F_1) \cdot P(F_2)$
- $P(F_1 \leftrightarrow F_2) = 1 + P(F_1) + P(F_2)$

En resumen, consiste en hacer corresponder las fórmulas falsas con el valor cero y las verdaderas con el uno. Por ejemplo, si una fórmula $(p_1 \wedge p_2)$ dada una valoración ($p_1 = \text{True}, p_2 = \text{True}$) es verdadera, su correspondiente polinomio $(x_1 * x_2)$ teniendo en cuenta la interpretación ($x_1 = 1, x_2 = 1$) debe valer uno ($1 + 1 =_{\mathbb{F}_2} 1$).

La implementación se hará en el módulo Transformaciones:

```
module Transformaciones where

import Logica
import Haskell4Maths (Vect(...))
```

```

        , var
        , vars
        , indices
        , eval
        , linear
        , (%%))

import F2 (PolF2)

import Test.QuickCheck

```

La función encargada de hacer dicha traducción es la función `tr.`, que equivale al mapeo P descrito anteriormente. Ésta recibe una fórmula proposicional del tipo `FProp` y devuelve un polinomio con coeficientes en \mathbb{F}_2 , es decir, del tipo `PolF2`.

```

-- | Por ejemplo,
--
-- >>> let [p1,p2] = [Atom "p1",Atom "p2"]
-- >>> tr p1
-- x1
-- >>> tr (p1 ∧ p2)
-- x1x2
-- >>> tr (p ∧ (q ∨ r))
-- qrx+qx+rx
tr :: FProp -> PolF2
tr T      = 1
tr F      = 0
tr (Atom ('p':xs)) = var ('x':xs)
tr (Atom xs)      = var xs
tr (Neg a)        = 1 + tr a
tr (Conj a b)     = tr a * tr b
tr (Disj a b)     = a' + b' + a' * b'
                  where a' = tr a
                        b' = tr b
tr (Impl a b)     = 1 + a' + a' * tr b
                  where a' = tr a
tr (Equi a b)     = 1 + tr a + tr b

```

Para la transformación contraria (de polinomios a fórmulas) se usará la función $\Theta : \mathbb{F}_2[x] \rightarrow \text{Form}(\mathcal{L})$ definida por:

- $\Theta(0) = \perp$
- $\Theta(1) = \top$
- $\Theta(x_i) = p_i$

- $\Theta(a + b) = \neg(\Theta(a) \leftrightarrow \Theta(b))$
- $\Theta(a \cdot b) = \Theta(a) \wedge \Theta(b)$

La función `(theta p)` transforma el polinomio `p` en la fórmula proposicional que le corresponde según la definición anterior.

```
-- | Por ejemplo,
--
-- >>> let [x1,x2] = [var "x1", var "x2"] :: [PolF2]
-- >>> theta 0
-- ⊥
-- >>> theta (x1*x2)
-- (p1 ∧ p2)
-- >>> theta (x1 + x2 + 1)
-- ¬(p1 ↔ ¬(p2 ↔ ⊤))

theta :: PolF2 -> FProp
theta 0      = F
theta 1      = T
theta (V [m]) = (theta' . mindices . fst) m
theta (V (x:xs)) = no (((theta' . mindices . fst) x) ↔ (theta (V xs)))

theta' :: [(String, t)] -> FProp
theta' []      = T
theta' [((x':v),i)] = Atom ('p':v)
theta' [((x':v),i):vs] = Conj (Atom ('p':v)) (theta' vs)
theta' [(v,i)]      = Atom v
theta' [(v,i):vs]   = Conj (Atom v) (theta' vs)
```

A continuación se definen dos propiedades que deben cumplir las funciones `tr` y `theta`.

Proposición 2.2.2. Sea f una fórmula proposicional cualquiera, $\Theta(P(f))$ es equivalente a f . La implementación de esta propiedad es:

```
-- |
-- >>> quickCheckWith (stdArgs {maxSize = 50}) prop_theta_tr
-- +++ OK, passed 100 tests.
prop_theta_tr :: FProp -> Bool
prop_theta_tr f = equivalentes (theta (tr f)) f
```

Notar que a la hora de chequear la propiedad anterior se ha acotado el tamaño máximo de las fórmulas proposicionales ya que en caso contrario se demora demasiado en ejecutarse.

Se define ahora la propiedad inversa:

Proposición 2.2.3. Sea p un polinomio de $\mathbb{F}_2[x]$, $P(\Theta(p)) = p$. Cuya implementación es:

```
prop_tr_theta :: PolF2 -> Bool
prop_tr_theta p = tr (theta p) == p
```

Sin embargo, al ejecutarlo nos devuelve Failed:

```
-- >>> quickCheck prop_tr_theta
-- *** Failed! Falsifiable (after 1 test):
-- x29^3x87^5+x30x74^2x80^4+x38^5x62^2
```

Esto se debe a los exponentes, que se pierden al transformar el polinomio en una fórmula proposicional. Por tanto, al reescribir el polinomio, éste es idéntico pero sin exponentes. Se tratará esto en la siguiente subsección y se comprobará que realmente ambos polinomios son iguales al estar en $\mathbb{F}_2[x]$.

2.2.5. Correspondencia entre valoraciones y puntos en \mathbb{F}_2^n

El comportamiento similar como funciones de la fórmula F y su traducción polinomial $P(F)$ son la base entre la semántica y las funciones polinomiales. Con idea de esclarecer qué se quiere decir con esto se explicará qué quiere decir *un comportamiento similar*:

- *De valoraciones a puntos*: Dada una valoración o interpretación $v : \mathcal{L} \rightarrow \{0, 1\}$ el valor de verdad de F respecto de v coincide con el valor de $P(F)$ en el punto $o_v \in \mathbb{F}_2^n$ definido por los valores dados por v ; $(o_v)_i = v(p_i)$. Es decir, para cada fórmula $F \in \text{Form}(\mathcal{L})$,

$$v(F) = P(F)((o_v)_1, \dots, (o_v)_n)$$

- *De puntos a valoraciones*: Cada $o = (o_1 \dots o_n) \in \mathbb{F}_2^n$ define una valoración v_o de la siguiente forma:

$$v_o(p_i) = 1 \text{ si y sólo si } o_i = 1$$

Entonces,

$$v_o \models F \Leftrightarrow P(F)(o_v) + 1 = 0 \Leftrightarrow o_v \in \mathcal{V}(1 + P(F))$$

donde $V(\cdot)$ se define como: Dado $a(\mathbf{x}) \in \mathbb{F}_2[x]$,

$$V(a(\mathbf{x})) = \{o \in \mathbb{F}_2^n : a(o) = 0\}$$

Por consiguiente, hay dos mapeos entre el conjunto de interpretaciones o valoraciones y los puntos de \mathbb{F}_2^n , que definen biyecciones entre modelos de F y puntos de la variedad algebraica $\mathcal{V}(1 + P(F))$;

$$\begin{array}{ccc} \text{Mod}(F) & \rightarrow & \mathcal{V}(1 + P(F)) \\ v & \rightarrow & o_v \end{array} \quad \begin{array}{ccc} \mathcal{V}(1 + P(F)) & \rightarrow & \text{Mod}(F) \\ o & \rightarrow & v_o \end{array}$$

Por ejemplo, sea la fórmula $F = p_1 \rightarrow p_2 \wedge p_3$. El polinomio asociado es $P(F) = 1 + x_1 + x_1x_2x_3$. La valoración $v = \{(p_1, 0), (p_2, 1), (p_3, 0)\}$ es modelo de F e induce el punto $o_v = (0, 1, 0) \in \mathbb{F}_2^3$ que a su vez pertenece a $\mathcal{V}(1 + P(F)) = \mathcal{V}(x_1 + x_1x_2x_3)$.

```
-- |
-- >>> let [p1,p2,p3] = map Atom ["p1","p2","p3"]
-- >>> let f = p1 → p2 ∧ p3
-- >>> tr f
-- x1x2x3+x1+1
-- >>> esModeloFormula [p3] f
-- True
-- >>> eval (1+(tr f)) [(var "x1",0),(var "x2",1),(var "x3",0)]
-- 0
```

2.2.6. Proyección polinomial

Consideremos ahora la parte derecha de la figura 2.1. Para simplificar la relación entre la semántica de la lógica proposicional y la geometría sobre cuerpos finitos se usará el mapa:

$$\begin{aligned} \Phi : \mathbb{F}_2[\mathbf{x}] &\rightarrow \mathbb{F}_2[\mathbf{x}] \\ \Phi\left(\sum_{\alpha \in I} \mathbf{x}^\alpha\right) &:= \sum_{\alpha \in I} \mathbf{x}^{sg(\alpha)} \end{aligned}$$

siendo $sg(\alpha) := (\delta_1, \dots, \delta_n)$ donde δ_i es 0 si $\alpha_i = 0$ y 1 en cualquier otro caso.

En la librería `HaskellForMaths` ya existe una función que calcula el representante de un polinomio en el grupo cociente por un ideal. Esta es la función `(%%)`. Sin embargo, ya que la búsqueda de la eficiencia es una máxima en este trabajo, se aprovechará el hecho de que calcular dicho representante equivale a reemplazar cada ocurrencia de x_i^k (con $k \in \mathbb{N}$) por x_i .

La función `(phi p)` calcula el representante de menor grado del polinomio p en el grupo cociente $\mathbb{F}_2[\mathbf{x}]/\mathbb{I}_2$, siendo $\mathbb{I}_2 = \{x_1 + x_1^2, \dots, x_n + x_n^2\}$ y $n \in \mathbb{N}$ el número total de variables.


```
-- | Por ejemplo,
-- >>> let [x1,x2] = [var "x1", var "x2"] :: [PolF2]
-- >>> phi (1+x1+x1^2*x2)
-- x1x2+x1+1
phi :: PolF2 -> PolF2
phi = linear (\m -> product [ var x | (x,i) <- mindices m])
```

Para poder comprobar la propiedad clave que justifica la redefinición de phi, es necesaria la función (ideal p) que devuelve el ideal (con menos generadores) respecto al cual se calcula el grupo cociente para buscar el representante.

```
-- | Por ejemplo,
--
-- >>> let [x1,x2] = [var "x1", var "x2"] :: [PolF2]
-- >>> ideal (1+x1+x1^2*x2)
-- [x1^2+x1,x2^2+x2]
ideal :: PolF2 -> [PolF2]
ideal p = [v+v^2 | v<-vars p]
```

La propiedad implementada queda:

```
-- |
-- >>> quickCheck prop_phi
-- +++ OK, passed 100 tests.
prop_phi :: PolF2 -> Bool
prop_phi p = phi p == p %% (ideal p)
```

Tal y como se ha descrito anteriormente, Φ selecciona un representante de la clase de equivalencia de $\mathbb{F}_2[x]/\mathbb{I}_2$. Dicho representante resulta ser también un polinomio, por lo que cuando se quiere asociar un polinomio a una fórmula proposicional basta aplicar la composición $\pi := \Phi \circ P$, que se llamará *proyección polinomial*.

Esta proyección es muy útil para manejar los polinomios ya que los simplifica en gran medida. Por ejemplo, sea $F = p_1 \rightarrow p_1 \wedge p_2$, entonces $P(F) = 1 + x_1 + x_1^2 x_2$ mientras que $\pi(F) = 1 + x_1 + x_1 x_2$.

La función `proyeccion p` es la implementación de la función $\pi(p)$:

```
-- | Por ejemplo,
-- >>> let [p1,p2] = [Atom "p1",Atom "p2"]
-- >>> proyeccion p1
```

```
-- x1
-- >>> tr (p1 → p1 ∧ p2)
-- x1^2x2+x1+1
-- >>> proyeccion (p1 → p1 ∧ p2)
-- x1x2+x1+1
proyeccion :: FProp -> PolF2
proyeccion = (phi . tr)
```

Conviene comprobar si se verifica que cualquier fórmula f es equivalente a $\theta(\pi(f))$:

```
-- |
-- >>> quickCheckWith (stdArgs {maxSize = 50}) prop_theta_proyeccion
-- +++ OK, passed 100 tests.
prop_theta_proyeccion :: FProp -> Bool
prop_theta_proyeccion f = equivalentes (theta (proyeccion f)) f
```

Además, como se ha solucionado el problema de los exponentes se puede comprobar la propiedad recíproca:

```
-- |
-- >>> quickCheck prop_proyeccion_theta
-- +++ OK, passed 100 tests.
prop_proyeccion_theta :: PolF2 -> Bool
prop_proyeccion_theta p = phi p == (proyeccion . theta) p
```

2.2.7. Bases de conocimiento e ideales

En esta subsección se recordará la correspondencia entre conjuntos algebraicos e ideales polinomiales (enfocado al cuerpo de coeficientes \mathbb{F}_2) y la lógica proposicional.

Definición 2.2.4. Dado un subconjunto $X \subseteq (\mathbb{F}_2)^n$, se denota por $I(X)$ al ideal de polinomios de $\mathbb{F}_2[x]$ que se anulan en X :

$$I(X) = \{a(\mathbf{x}) \in \mathbb{F}_2[x] : a(u) = 0 \text{ para cualquier } u \in X\}$$

Simétricamente, a partir de un subconjunto $J \subseteq \mathbb{F}_2[x]$ es posible definir el conjunto algebraico $\mathcal{V}(J)$ comentado anteriormente:

$$\mathcal{V}(J) = \{u \in (\mathbb{F}_2)^n : a(u) = 0 \text{ para cualquier } a(\mathbf{x}) \in J\}$$

Antes de enunciar y demostrar el teorema de Nullstellensatz para cuerpos finitos [1] (concretamente \mathbb{F}_2) es necesario un lema:

Lema 2.2.5. Sea un polinomio $p \in \mathbb{F}_2[\mathbf{x}]$, entonces $p \in \mathbb{I}_2^n \Leftrightarrow p(\mathbf{z}) = 0 \forall \mathbf{z} \in (\mathbb{F}_2)^n$

Prueba: La implicación hacia la derecha es trivial ya que si $p \in \mathbb{I}_2^n$ entonces $p \in \langle x_1^2 + x_1, \dots, x_n^2 + x_n \rangle$, es decir, $p = \sum_{i=1}^n q_i(x_i^2 + x_i)$ donde $q_i \in \mathbb{F}_2[\mathbf{x}]$ con $i = 1, \dots, n$. Como todos los $(x_i^2 + x_i)$ se anulan en todo punto de $\mathbb{F}_2[\mathbf{x}]$ entonces p también.

La implicación hacia la izquierda se probará por inducción en el número de variables. En el caso de una única variable ($n = 1$), la división euclídea de p por $\mathbb{I}_2^1 = x_1^2 + x_1$ queda $p = a * (x_1^2 + x_1 + b)$ con $b = b_0 + b_1 x_1$. De la hipótesis tenemos que $b(0) = b(1) = 0$, luego $b = 0$ y por tanto $p \in \mathbb{I}_2^1$.

Sea $n \geq 1$, se usa también la división respecto \mathbb{I}_1 por lo que $p = a * \mathbb{I}_2^1 + b$ donde $b = b_0 + b_1 x_1$; $b_0, b_1 \in \mathbb{F}_2[x_2, \dots, x_n]$. Fijando un punto cualquiera $\mathbf{z} = (z_2, \dots, z_n) \in \mathbb{F}_2^{n-1}$, el polinomio b respecto de la variable x_1 queda: $b_0(\mathbf{z}) + b_1(\mathbf{z})x_1 = 0$ para $x_1 = 0$ y para $x_1 = 1$. Como, $b(\mathbf{z})(x_1)$ es de grado 1 y tiene 2 raíces, entonces $b_0(\mathbf{z}) = b_1(\mathbf{z}) = 0$. Aplicando la hipótesis de inducción $b_0, b_1 \in \langle x_2^2 + x_2, \dots, x_n^2 + x_n \rangle$ luego $p \in \mathbb{I}_2^n$. \square

Teorema 2.2.6. Teorema de Nullstellensatz con coeficientes en el cuerpo \mathbb{F}_2

1. Si $A \subseteq (\mathbb{F}_2)^n$, entonces $\mathcal{V}(I(A)) = A$
2. Para todo $\mathfrak{J} \in \text{Ideales}(\mathbb{F}_2[\mathbf{x}])$, $I(\mathcal{V}(\mathfrak{J})) = \mathfrak{J} + \mathbb{I}_2$

Prueba:

1. Se prueba por doble contención. La primera ($A \subseteq \mathcal{V}(I(A))$) es trivial: sea $a \in A$ entonces por definición de I se tiene que $\forall p \in I(A)$, $p(a) = 0$; y, por lo tanto, $a \in \mathcal{V}(I(A))$.

La contención contraria ($\mathcal{V}(I(A)) \subseteq A$) se probará por *reductio ad absurdum*. Supongamos que existe un punto $a \in \mathcal{V}(I(A))$ pero que $a \notin A$. Sea el polinomio $p_A(\mathbf{x}) = 1 + \sum_{\alpha \in A} \prod_{i=1}^n (x_i + \alpha_i + 1)$. Es fácil ver que $p_A(u) = 0$ si $u \in A$. De la hipótesis se tiene que $p_A(a) \neq 0$, y de la definición de p_A que $p_A \in I(A)$. De esto se deduce que $a \notin \mathcal{V}(I(A))$, lo cual es una contradicción. \square

2. Del teorema de las bases de Hilbert se deduce que existen $j_1, \dots, j_s \in \mathbb{F}_2[\mathbf{x}]$ tales que $\mathfrak{J} = \langle j_1, \dots, j_s \rangle$. Entonces, la prueba de la contención hacia la izquierda, $\mathfrak{J} + \mathbb{I}_2 \subseteq I(\mathcal{V}(\mathfrak{J}))$, es inmediata porque todos los polinomios j_k e $i_{k'}$ con $1 \leq k \leq s$, $1 \leq k' \leq n$ se anulan en $\mathcal{V}(\mathfrak{J})$.

Para probar la contención inversa se fijan el polinomio $p \in I(\mathcal{V}(\mathfrak{J}))$ y el subconjunto $A \subseteq (\mathbb{F}_2)^n$ definido como $A := (\mathbb{F}_2)^n \setminus \mathcal{V}(\mathfrak{J})$. Además se tiene que para todo $a = (a_1, \dots, a_n) \in A$ existe un índice i_a tal que $j_{i_a} \neq 0$. Entonces, el polinomio $g = p \cdot (\prod_{a \in A} (j_{i_a} - j_{i_a}(a)))$ se anula en todo $(\mathbb{F}_2)^n$ (porque p se anula en $\mathcal{V}(\mathfrak{J})$ y $\prod_{a \in A} (j_{i_a} - j_{i_a}(a))$ en A). Por el lema 2.2.5, $g \in \mathbb{I}_2^n$. Desarrollando el producto en g se puede escribir el polinomio como $g = bp + h$, donde $h \in \mathfrak{J}$ y $b =$

$\prod_{a \in A} (-j_{i_a}(a))$. Por tanto, $bp = g - h \in \mathfrak{J} + \mathbb{I}_2^n$ y $b \neq 0$. De esto se sigue que $p \in \mathfrak{J} + \mathbb{I}_2^n$. \square

Del teorema de Nullstellensatz se sigue que:

$$F \equiv F' \text{ si y sólo si } P(F) = P(F') \pmod{\mathbb{I}_2}$$

Por consiguiente, $F \equiv F'$ si y sólo si $\pi(F) = \pi(F')$. Para la prueba del Teorema 2.2.8 es necesario el siguiente lema:

Lema 2.2.7. Sean los polinomios $R, P_1, \dots, P_m \in \mathbb{F}_2[\mathbf{x}]$, sea el ideal $\mathfrak{J} = \langle P_1, \dots, P_m \rangle$ y sea $\mathcal{R} = \mathfrak{J} + \mathbb{I}_2^n$. Entonces,

$$R \in \mathcal{R} \iff R(a) = 0, \forall a \in A = \{z \in (\mathbb{F}_2)^n : P_1(z) = \dots = P_m(z) = 0\}$$

Prueba: La implicación hacia la derecha (\Rightarrow) es trivial ya que $P_i(a) = 0$ y $a_j^2 + a_j = 0$ para todo $i = 1, \dots, m$ y $j = 1, \dots, n$. Para la otra implicación (\Leftarrow), se define el conjunto $B = \mathbb{F}_2^n \setminus A$. Entonces $\forall z \in B$ existe un índice $i_z \in \{1, \dots, m\}$ tal que $P_{i_z}(z) \neq 0$. Se define el polinomio $S = R \cdot \prod_{z \in B} (P_{i_z} - P_{i_z}(z))$. Notar que este polinomio se anula en todo $(\mathbb{F}_2)^n$ ya que es producto de R , que se anula en A ; y de $\prod_{z \in B} (P_{i_z} - P_{i_z}(z))$, que se anula en B . El lema 2.2.5 implica que $S \in \mathbb{I}_2^n$. Si se reescribe S desarrollando el producto: $S = b \cdot R + P'$, con $P' \in \mathfrak{J}$ y $b = \prod_{z \in B} (-P_{i_z}(z)) \in \mathbb{F}_2$; se deduce que $b \cdot R = S - P' \in \mathfrak{J} + \mathbb{I}_2^n = \mathcal{R}$. Finalmente, como $b \neq 0$, se tiene que $R \in \mathcal{R}$. \square

El siguiente teorema resume la relación entre la lógica proposicional y $\mathbb{F}_2[\mathbf{x}]$:

Teorema 2.2.8. Sea $K = \{F_1, \dots, F_m\}$ un conjunto de fórmulas proposicionales y G una fórmula proposicional. Las siguientes sentencias son equivalentes:

1. $\{F_1, \dots, F_m\} \models G$
2. $1 + P(G) \in \langle 1 + P(F_1), \dots, 1 + P(F_m) \rangle + \mathbb{I}_2$
3. $\mathcal{V}(1 + P(F_1), \dots, 1 + P(F_m)) \subseteq \mathcal{V}(1 + P(G))$

Prueba: La estructura que se seguirá es ver que la segunda condición se cumple si y sólo si las otras dos lo hacen. Por el lema 2.2.7:

$$1 + P(G) \in \langle 1 + P(F_1), \dots, 1 + P(F_m) \rangle + \mathbb{I}_2 \iff$$

$$\iff 1 + P(G)(a) = 0 \quad \forall a \in A = \{z \in (\mathbb{F}_2)^n : 1 + P(F_1)(z) = \dots = 1 + P(F_m)(z) = 0\}$$

En otras palabras, si se anulan todos los $(1 + P(F_i))$, $i = 1, \dots, m$, entonces se anula $(1 + P(G))$. Esto pasa si y sólo si:

$$\mathcal{V}(1 + P(F_1), \dots, 1 + P(F_m)) \subseteq \mathcal{V}(1 + P(G))$$

quedando así probado ($2 \Leftrightarrow 3$).

Además, es fácil ver que $A = \{o_v : v \in \text{Mod}(\{F_1, \dots, F_m\})\}$, y por tanto,

$$\begin{aligned} 1 + P(G)(a) = 0 \quad \forall a \in A &\Leftrightarrow P(G)(a) = 1 \quad \forall a \in A \Leftrightarrow A \subseteq \{o'_v : v \in \text{Mod}(G)\} \Leftrightarrow \\ &\Leftrightarrow \text{Mod}(\{F_1, \dots, F_m\}) \subseteq \text{Mod}(G) \Leftrightarrow \{F_1, \dots, F_m\} \models G \end{aligned}$$

quedando así probado ($2 \Leftrightarrow 1$). □

Se sabe que todo conjunto $X \subseteq (\mathbb{F}_2)^n$ es un conjunto algebraico; de hecho, existen $a_X \in \mathbb{F}_2[\mathbf{x}]$ tal que $\mathcal{V}(a_X) = X$. Por lo que, aplicando el teorema de Nullstellensatz se tiene que $I(\mathcal{V}(a_X)) = (a_X) + \mathbb{I}_2$, de lo que se sigue que el anillo de coordenadas de X como variedad algebraica es:

$$\mathbb{F}_2[\mathbf{x}] / I(X) \cong (\mathbb{F}_2[\mathbf{x}] / (a_X)) / \mathbb{I}_2$$

Cualquier ideal J_X tal que $\mathcal{V}(J_X) = X$ se puede usar para describir el anillo de coordenadas. Por consiguiente y con el objetivo de simplificar la notación, se asumirá que $\mathbb{I}_2 \subseteq J_X$ si es necesario. De manera similar, dada una base de conocimiento K , se define el ideal:

$$J_K = (\{1 + P(F) : F \in K\})$$

y entonces

$$v \models K \Leftrightarrow o_v \in \mathcal{V}(J_K)$$

Definición 2.2.9. El *anillo de coordenadas* de K se define como el correspondiente a la variedad algebraica $V(J_K)$, que, por el teorema de Nullstellensatz, es $\mathbb{F}_2[\mathbf{x}] / (J_K) / \mathbb{I}_2$.

Capítulo 3

Regla de independencia y prueba no clausal de teoremas

En el presente capítulo se expondrá el diseño de un método de prueba de teoremas basado en la consistencia o inconsistencia del conjunto de fórmulas de partida, así como de la negación de la fórmula que se quiere deducir. En definitiva se basa en el hecho de que, si K es una base de conocimiento y F una fórmula proposicional:

$$K \models F \text{ si y sólo si } K \cup \{\neg F\} \text{ es inconsistente}$$

La idea principal será hallar la inconsistencia del conjunto de fórmulas mediante la saturación de dicho conjunto en la constante \perp . Para saturar, se usarán las ya mencionadas retracciones conservativas, que se calcularán mediante lo que se denominará un *operador de omisión de variables*.

Este operador eliminará una de las variables cada vez que se aplique, obteniendo a cada paso un conjunto equivalente de fórmulas en los que se usa una variable menos. Finalmente, como hay un número finito de variables en $K \cup \{\neg F\}$, llegará un momento en el que no queden variables y que sólo queden las constantes \top o \perp (sólo una de ellas), habiendo así refutado o probado, respectivamente, el teorema.

Un cambio relevante en la estructura de este capítulo respecto a la seguida en el capítulo anterior es que se expondrán en primer lugar las bases teóricas que fundamentan el modelo lógico-algebraico de razonamiento; y, posteriormente, se implementará en Haskell dicho modelo.

3.1. Retracción conservativa mediante omisión de variables

En esta sección se presenta cómo calcular retracciones conservativas usando los ya mencionados operadores *de omisión* (o *de olvido*). Dichos operadores son mapas del tipo:

$$\delta : \text{Form}(\mathcal{L}) \times \text{Form}(\mathcal{L}) \longrightarrow \text{Form}(\mathcal{L})$$

donde 2^X representa al conjunto potencia de X .

Definición 3.1.1. Sea δ un operador: $\delta : \text{Form}(\mathcal{L}) \times \text{Form}(\mathcal{L}) \longrightarrow \text{Form}(\mathcal{L} \setminus \{p\})$ se dice que es:

1. *robusto* si $\{F, G\} \models \delta(F, G)$.
2. un *operador de omisión* para la variable $p \in \mathcal{L}$ si:

$$\delta(F, G) \equiv [\{F, G\}, \mathcal{L} \setminus \{p\}]$$

Una caracterización muy útil de los operadores se puede deducir de la siguiente propiedad semántica: Si δ es un operador de omisión, los modelos de $\delta(F, G)$ son precisamente las *proyecciones* de los modelos de $\{F, G\}$ (ver figura 3.1).




Figura 3.1: Interpretación semántica del operador de omisión (Lema de elevación)

Lema 3.1.2. (Lema de elevación) Sean $v : \mathcal{L} \setminus \{p\} \rightarrow \{0, 1\}$ una valoración o interpretación, $F, G \in \text{Form}(\mathcal{L})$ fórmulas y δ un operador de omisión de la variable p . Las siguientes condiciones son equivalentes:

1. $v \models \delta(F, G)$
2. Existe una valoración $\hat{v} : \mathcal{L} \rightarrow \{0, 1\}$ tal que $\hat{v} \models F \wedge G$ y $\hat{v} \upharpoonright_{\mathcal{L} \setminus \{p\}} = v$

Prueba: ($1 \Rightarrow 2$): Dada una valoración v , se considera la fórmula

$$H_v = \bigwedge_{q \in \mathcal{L} \setminus \{p\}} q^v$$

donde q^v es q si $v(q) = 1$ y $\neg q$ en otro caso. Es claro que v es la única valoración de $\mathcal{L} \setminus \{p\}$ que es modelo de H_v .

Supongamos que existe $v : \mathcal{L} \setminus \{p\} \rightarrow \{0, 1\}$ modelo de $\delta(F, G)$, pero que no se puede extender a un modelo de $F \wedge G$. Entonces la fórmula

$$H_v \rightarrow \neg(F \wedge G)$$

es una tautología, en particular:

$$\{F, G\} \models H_v \rightarrow \neg(F \wedge G)$$

Como $\{F, G\} \models F \wedge G$, usando *modus tollens* se tiene $\{F, G\} \models \neg H_v$. Así que $\delta(F, G) \models \neg H_v$, por ser δ una retracción conservativa. Este hecho es una contradicción porque $v \models \delta(F, G) \wedge H_v$.

($2 \Rightarrow 1$): La extensión \hat{v} verifica que:

$$\hat{v} \models F \wedge G \models [\{F, G\}, \mathcal{L} \setminus \{p\}] \models \delta(F, G)$$

Como $\delta(F, G) \in \text{Form}(\mathcal{L} \setminus \{p\})$, la valoración $v = \hat{v} \upharpoonright_{\mathcal{L} \setminus \{p\}}$ también es modelo de $\delta(F, G)$. \square

En particular, el resultado es cierto para la propia retracción conservativa canónica $[K, \mathcal{L} \setminus \{p\}]$, porque si consideramos la fórmula $\bigwedge K := \bigwedge_{F \in K} F$,

$$[K, \mathcal{L} \setminus \{p\}] \equiv \delta_p(\bigwedge K, \bigwedge K)$$

Un caso interesante aparece cuando $\delta_p(F_1, F_2) \equiv \top$. En este caso, toda valoración parcial en $\mathcal{L} \setminus \{p\}$ se puede extender a un modelo de $\{F_1, F_2\}$.

La siguiente caracterización será útil más adelante:

Corolario 3.1.3. Sea $\delta : \text{Form}(\mathcal{L}) \times \text{Form}(\mathcal{L}) \longrightarrow \text{Form}(\mathcal{L} \setminus \{p\})$ un operador robusto. Las siguientes condiciones son equivalentes:

1. δ es un operador de omisión de la variable p .
2. Para cualesquiera $F, G \in \text{Form}(\mathcal{L})$ y $v \models \delta(F, G)$ valoración sobre $\mathcal{L} \setminus \{p\}$, existe una extensión de v modelo de $\{F, G\}$.

Prueba: ($1 \Rightarrow 2$): Cierta por el Lema de Elevación (Lema 3.1.2).

($2 \Rightarrow 1$): Sean F y G dos fórmulas. Como δ es robusto, basta probar que:

$$\delta(F, G) \models [\{F, G\}, \mathcal{L} \setminus \{p\}]$$

Supongamos que no es cierto. En ese caso, existe una fórmula $H \in \text{Form}(\mathcal{L} \setminus \{p\})$ tal que $[\{F, G\}, \mathcal{L} \setminus \{p\}] \models H$ (luego H también es consecuencia lógica de $\{F, G\}$), pero existe una valoración v que satisface $v \models \delta(F, G) \wedge \neg H$. Por (2), existe \hat{v} extensión de v que es modelo de $\{F, G\}$. Por tanto, $\{F, G\} \models \neg H$, lo que es una contradicción. \square

Corolario 3.1.4. Si $p \notin \text{var}(F)$, y δ_p es un operador de omisión de p , entonces

$$\delta_p(F, F) \equiv F \quad \text{y} \quad \delta_p(F, G) \equiv \{F, \delta_p(G, G)\}$$

Prueba: Si $p \notin \text{var}(F)$, entonces $\{F\} \equiv [\{F\}, \mathcal{L} \setminus \{p\}] \equiv \delta_p(F, F)$.

Por otro lado, $\delta_p(F, G) \equiv [\{F, G\}, \mathcal{L} \setminus \{p\}] \models \{F, \delta_p(G, G)\}$. Para probar que en realidad se trata de una equivalencia se mostrara que tienen los mismos modelos.

Sea v una valoración sobre $\mathcal{L} \setminus \{p\}$ tal que $v \models \{F, \delta_p(G, G)\}$. Entonces existe \hat{v} (una extensión de v) tal que $\hat{v} \models G$. Como $\hat{v} \models F$ se tiene por el Lema de Elevación que $v \models \delta(F, G)$. \square

Retracciones conservativas inducidas por un operador de omisión

En el artículo [5] J. Lang et al. presentan un método de omisión de X (un conjunto de variables de la fórmula F), denotado por $\text{forget}(F, X)$ y basado en la construcción de disyunciones de la siguiente forma:

$$\begin{aligned} \text{forget}(F, \emptyset) &= F \\ \text{forget}(F, \{x\}) &= F\{x/\top\} \cup F\{x/\perp\} \\ \text{forget}(F, \{x\} \cup Y) &= \text{forget}(\text{forget}(F, Y), \{x\}) \end{aligned}$$

Notar que con este método el tamaño de $\text{forget}(F, Y)$ puede ser realmente grande. En el método que se expone en el trabajo se pretende simplificar la representación mediante el uso de operaciones algebraicas sobre proyecciones polinomiales.

Tal y como se ha descrito anteriormente, el operador de omisión de la variable p actúa entre pares de fórmulas. A continuación, se extenderá la definición del operador de forma que se pueda aplicar a conjuntos de fórmulas o bases de conocimiento.

Definición 3.1.5. Sea δ_p un operador de omisión de la variable p y K una base de conocimiento. Se define $\delta_p[\cdot]$ como:

$$\begin{aligned}\delta_p[\cdot] &: 2^{Form(\mathcal{L})} \rightarrow 2^{Form(\mathcal{L})} \\ \delta_p[K] &:= \{\delta_p(F, G) : F, G \in K\}\end{aligned}$$

Si se supone que se tiene un operador de omisión δ_p para cada $p \in \mathcal{L}$:

Definición 3.1.6. Se llamará *saturación* de la base de conocimiento K al proceso de aplicar los operadores $\delta_p[\cdot]$ (en algún orden) respecto a todas las variables proposicionales de $\mathcal{L}(K)$, denotando al resultado como $sat_\delta(K)$ (el cual será un subconjunto de $\{\top, \perp\}$).

Posteriormente se verá que $sat_\delta(K)$ no depende del orden de aplicación de los operadores. Además, se probará que debido a que los operadores de omisión son robustos, si K es consistente entonces necesariamente $sat_\delta(K) = \{\top\}$.

A partir de los operadores de omisión resulta natural definir el siguiente cálculo lógico:

Definición 3.1.7. Sea K una base de conocimiento, $F \in Form(\mathcal{L})$ y $\{\delta_p : p \in \mathcal{L}(K)\}$ una familia de operadores de omisión.

- $A \vdash_\delta$ -prueba en K es una secuencia de fórmula F_1, \dots, F_n tal que para todo $i \leq n$, $F_i \in K$ ó existen $F_j, F_k (j, k < i)$ tal que $F_i = \delta_p(F_j, F_k)$ para algún $p \in \mathcal{L}$.
- $K \vdash_\delta F$ si existe una \vdash_δ -prueba en K , F_1, \dots, F_n , con $F_n = F$.
- Una \vdash_δ -refutación es una \vdash_δ -prueba de \perp .

La completitud (refutacional) del cálculo asociado a los operadores de omisión se enuncia como sigue:

Teorema 3.1.8. Sea $\{\delta_p : p \in \mathcal{L}\}$ una familia de operadores de omisión. Entonces \vdash_δ es refutacionalmente completo, es decir, K es inconsistente si y sólo si $K \vdash_\delta \perp$.

Prueba: La idea es saturar la base de conocimiento como en la Figura (3.2). Si $sat_\delta(K) = \{\top\}$, entonces, aplicando repetidas veces el lema de elevación, se puede extender la valoración vacía (la cual es modelo de $\{\top\}$) a un modelo de K .

Si $\perp \in sat_\delta(K)$ entonces K es inconsistente, porque $K \models sat_\delta(K)$ por robustez de los operadores de omisión. \square




Figura 3.2: Decidir la consistencia usando operadores de omisión (∂_{p_i})

Corolario 3.1.9. $\delta_p[K] \equiv [K, \mathcal{L} \setminus \{p\}]$

Prueba: Por robustez del operador de omisión δ_p se tiene:

$$[K, \mathcal{L} \setminus \{p\}] \models \delta_p[K]$$

Para probar la otra dirección, se considera $F \in [K, \mathcal{L} \setminus \{p\}]$ y se supone que $\delta_p[K] \not\models F$. Entonces $\delta_p[K] + \{\neg F\}$ es consistente. En particular, si se satura se tiene: $\text{sat}_\delta(\delta_p[K] \cup \{\neg F\}) = \{\top\}$.

Como $p \notin \text{var}(\neg F)$ se puede usar el corolario 3.1.4, obteniendo que para todo $G \in K$:

$$\delta_g(\neg F, G) \equiv \{\neg F, \delta_g(G, G)\} \quad \text{y} \quad \delta_p(\neg F, \neg F) \equiv \neg F$$

Por consiguiente,

$$\delta_p[K \cup \{\neg F\}] \equiv \delta_p[K] \cup \{\neg F\}$$

así que, aplicando saturación empezando por p :

$$\text{sat}_\delta(K \cup \{\neg F\}) \equiv \text{sat}_\delta(\delta_p[K] \cup \{\neg F\})$$

Lo que indica que $K \cup \{\neg F\}$ es consistente, luego $K \not\models F$, que es una contradicción. \square

Dados $Q \subseteq \mathcal{L}$ y un orden lineal $q_1 < \dots < q_k$ sobre Q , se define el operador:

$$\delta_{Q, <} := \delta_{q_1} \circ \dots \circ \delta_{q_k}$$

Este, no es más que la aplicación sucesiva del operador de omisión δ respecto de

cada una de las variables de Q en un orden dado. De hecho, si prescindimos de hacer explícito el orden, el operador está bien definido módulo equivalencia lógica. En otras palabras, dados dos órdenes sobre Q cualesquiera, producen bases de conocimiento equivalentes. Esto es cierto puesto que para cada $p, q \in \mathcal{L}$, usando el corolario 3.1.9 se sigue que:

$$\delta_p \circ \delta_q[K] \equiv \delta_q \circ \delta_p[K]$$

ya que ambas son equivalentes a $[K, \mathcal{L} \setminus \{p, q\}]$. Por consiguiente y con la idea de simplificar la notación, se escribirá δ_Q cuando no importe la presentación sintáctica de dicha base de conocimiento.

Como consecuencia del teorema 3.1.8 y el corolario 3.1.9 se da que el hecho de saber si una fórmula es o no consecuencia lógica de otra (o de una base de conocimiento) se puede reducir a otro problema similar en el que sólo se trabaje con las variables de la fórmula objetivo. Basta con eliminar de forma sucesiva todas las variables de la base de conocimiento que no ocurran en la fórmula objetivo. A esta propiedad se le conoce como *Propiedad de localización*:

Corolario 3.1.10. (Propiedad de Localización [3]) Las siguientes condiciones son equivalentes:

1. $K \models F$
2. $\delta_{\mathcal{L} \setminus \text{var}(F)}[K] \models F$

Prueba: Trivial, porque $\delta_{\mathcal{L} \setminus \text{var}(F)}[K] \equiv [K, \text{var}(F)]$

3.2. Derivadas Booleanas

Con el objetivo de definir el operador de omisión que se usará en la herramienta, se hará uso de las derivadas polinomiales aunque no de forma directa. En la práctica, se traducirá la derivada sobre $\mathbb{F}_2[x]$ tradicional a un operador sobre fórmulas proposicionales.

Antes de continuar se expondrán diversas propiedades básicas. Recaltar que una derivación sobre un anillo R es una mapa $d : R \rightarrow R$ verificando que $\forall a, b \in R$:

- $d(a + b) = d(a) + d(b)$
- $d(a \cdot b) = d(a) \cdot b + a \cdot d(b)$

La traducción al mundo de la lógica de las derivaciones se construye tal y como sigue:

Definición 3.2.1. [3] Un mapa $\partial : \text{Form}(\mathcal{L}) \rightarrow \text{Form}(\mathcal{L})$ es una *derivada Booleana* si existe una derivación d sobre $\mathbb{F}_2[\mathbf{x}]$ tal que el siguiente diagrama sea conmutativo:

$$\begin{array}{ccc} \text{Form}(\mathcal{L}) & \xrightarrow{\partial} & \text{Form}(\mathcal{L}) \\ \pi \downarrow & \# & \uparrow \Theta \\ \mathbb{F}_2[\mathbf{x}] & \xrightarrow{d} & \mathbb{F}_2[\mathbf{x}] \end{array}$$

O lo que es lo mismo:

$$\partial = \Theta \circ d \circ \pi$$

En este trabajo, la atención recae sobre una derivada booleana en particular, denotada por $\frac{\partial}{\partial p}$, e inducida por la derivada $d = \frac{\partial}{\partial x_p}$. El siguiente resultado muestra una expresión semánticamente equivalente de esta derivada:

Proposición 3.2.2. $\frac{\partial}{\partial x_p} F \equiv \neg(F\{p/\neg p\} \leftrightarrow F)$

Prueba: Es fácil ver que

$$\pi(F\{p/\neg p\})(x) = \pi(F)(x_1, \dots, x_p + 1, \dots, x_n)$$

Por otro lado, se cumple que para fórmulas polinomiales:

$$\frac{\partial}{\partial x} a(x) = a(x + 1) + a(x)$$

Así que,

$$\frac{\partial}{\partial x_p} \pi(F) = \pi(F)(x_1, \dots, x_p + 1, \dots, x_n) + \pi(F)(x_1, \dots, x_n)$$

Y por tanto, aplicando Θ se tiene

$$\frac{\partial}{\partial p} F = \Theta\left(\frac{\partial}{\partial x_p} \pi(F)\right) \equiv \neg(F\{p/\neg p\} \leftrightarrow F)$$

□

3.3. Regla de independencia

3.4. Cálculo lógico

Capítulo 4

Aplicaciones

En este capítulo se hace una breve introducción a la programación funcional en Haskell suficiente para entender su aplicación en los siguientes capítulos. Para una introducción más amplia se pueden consultar los apuntes de la asignatura de Informática de 1º del Grado en Matemáticas ([2]).

El contenido de este capítulo se encuentra en el módulo PFH

```
module PFH where
import Data.List
```

4.1. Introducción a Haskell

En esta sección se introducirán funciones básicas para la programación en Haskell. Como método didáctico, se empleará la definición de funciones ejemplos, así como la redefinición de funciones que Haskell ya tiene predefinidas, con el objetivo de que el lector aprenda *“a montar en bici, montando”*.

A continuación se muestra la definición (cuadrado x) es el cuadrado de x. Por ejemplo, La definición es

```
-- |
-- >>> cuadrado 3
-- 9
-- >>> cuadrado 4
-- 16
cuadrado :: Int -> Int
cuadrado x = x * x
```


Capítulo 5

Conclusión

En este capítulo se hace una breve introducción a la programación funcional en Haskell suficiente para entender su aplicación en los siguientes capítulos. Para una introducción más amplia se pueden consultar los apuntes de la asignatura de Informática de 1º del Grado en Matemáticas ([2]).

El contenido de este capítulo se encuentra en el módulo PFH

```
module PFH where
import Data.List
```

5.1. Introducción a Haskell

En esta sección se introducirán funciones básicas para la programación en Haskell. Como método didáctico, se empleará la definición de funciones ejemplos, así como la redefinición de funciones que Haskell ya tiene predefinidas, con el objetivo de que el lector aprenda *“a montar en bici, montando”*.

A continuación se muestra la definición (cuadrado x) es el cuadrado de x. Por ejemplo, La definición es

```
-- |
-- >>> cuadrado 3
-- 9
-- >>> cuadrado 4
-- 16
cuadrado :: Int -> Int
cuadrado x = x * x
```


Bibliografía

- [1] J. C. Agudelo-Agudelo, C. A. Agudelo-González, and O. E. García-Quintero. On polynomial semantics for propositional logics. *Journal of Applied Non-Classical Logics*, 26(2):103–125, 2016.
- [2] J. Alonso. [Temas de programación funcional](#). Technical report, Univ. de Sevilla, 2015.
- [3] G. A. Aranda-Corral, J. Borrego-Díaz, and M. M. Fernández-Lebrón. *Conservative Retractions of Propositional Logic Theories by Means of Boolean Derivatives: Theoretical Foundations*, pages 45–58. Springer Berlin Heidelberg, Berlin, Heidelberg, 2009.
- [4] D. Kapur and P. Narendran. An equational approach to theorem proving in first-order predicate calculus. *SIGSOFT Softw. Eng. Notes*, 10(4):63–66, Aug. 1985.
- [5] J. Lang, P. Liberatore, and P. Marquis. Propositional independence: Formula-variable independence and forgetting. *J. Artif. Int. Res.*, 18(1):391–443, May 2003.

Índice alfabético

FProp, 13
Interpretacion, 16
VarProp, 13
 \leftrightarrow , 15
 \rightarrow , 14
 \vee , 14
 \wedge , 14
cuadrado, 9, 53, 55
equivalentesKB, 23
equivalentes, 23
esConsecuenciaKB, 22
esConsecuencia, 21
esConsistente, 20
esInconsistente, 21
esInsatisfacible, 18
esModeloFormula, 16
esModeloKB, 19
esSatisfacible, 18
esValida, 18
interpretacionesForm, 17
interpretacionesKB, 19
modelosFormula, 17
modelosKB, 20
no, 14
significado, 16
simbolosPropForm, 17
simbolosPropKB, 19
sustituye, 15